

var svar — Structural vector autoregressive models
[Description](#)[Quick start](#)[Menu](#)[Syntax](#)[Options](#)[Remarks and examples](#)[Stored results](#)[Methods and formulas](#)[Acknowledgment](#)[References](#)[Also see](#)

Description

`svar` fits a vector autoregressive model subject to short- or long-run constraints you place on the resulting impulse–response functions (IRFs). Economic theory typically motivates the constraints, allowing a causal interpretation of the IRFs to be made. See [\[TS\]](#) [var intro](#) for a list of commands that are used in conjunction with `svar`.

Quick start

Structural VAR for `y1`, `y2`, and `y3` using `tsset` data with short-run constraints on impulse responses given by predefined matrices `A` and `B`

```
svar y1 y2 y3, aeq(A) beq(B)
```

Structural VAR for `y1`, `y2`, and `y3` with long-run constraint on impulse responses given by the predefined matrix `C`

```
svar y1 y2 y3, lreq(C)
```

Add exogenous variables `x1` and `x2`

```
svar y1 y2 y3, lreq(C) exog(x1 x2)
```

As above, but include third and fourth lags of the dependent variables instead of first and second

```
svar y1 y2 y3, lreq(C) exog(x1 x2) lags(3 4)
```

Menu

Statistics > Multivariate time series > Structural vector autoregression (SVAR)

Syntax

Short-run constraints

```
svar devarlist [if] [in], { aconstraints(constraintsa) aeq(matrixaeq)
acns(matrixacns) bconstraints(constraintsb) beq(matrixbeq) bcns(matrixbcns) }
[short_run_options]
```

Long-run constraints

```
svar devarlist [if] [in], { lrconstraints(constraintslr) lreq(matrixlreq)
lrcns(matrixlrcns) } [long_run_options]
```

2 var svar — Structural vector autoregressive models

<i>short_run_options</i>	Description
Model	
<u>noconstant</u>	suppress constant term
* <u>aconstraints</u> (<i>constraints_a</i>)	apply previously defined <i>constraints_a</i> to A
* <u>aeq</u> (<i>matrix_{aeq}</i>)	define and apply to A equality constraint matrix <i>matrix_{aeq}</i>
* <u>acns</u> (<i>matrix_{acns}</i>)	define and apply to A cross-parameter constraint matrix <i>matrix_{acns}</i>
* <u>bconstraints</u> (<i>constraints_b</i>)	apply previously defined <i>constraints_b</i> to B
* <u>beq</u> (<i>matrix_{beq}</i>)	define and apply to B equality constraint matrix <i>matrix_{beq}</i>
* <u>bcns</u> (<i>matrix_{bcns}</i>)	define and apply to B cross-parameter constraint <i>matrix_{bcns}</i>
<u>lags</u> (<i>numlist</i>)	use lags <i>numlist</i> in the underlying VAR
Model 2	
<u>exog</u> (<i>varlist_{exog}</i>)	use exogenous variables <i>varlist</i>
<u>varconstraints</u> (<i>constraints_v</i>)	apply <i>constraints_v</i> to underlying VAR
<u>noislog</u>	suppress SURE iteration log
<u>isiterate</u> (#)	set maximum number of iterations for SURE; default is <code>isiterate(1600)</code>
<u>istolerance</u> (#)	set convergence tolerance of SURE
<u>noisure</u>	use one-step SURE
<u>dfk</u>	make small-sample degrees-of-freedom adjustment
<u>small</u>	report small-sample <i>t</i> and <i>F</i> statistics
<u>noidencheck</u>	do not check for local identification
<u>nobigf</u>	do not compute parameter vector for coefficients implicitly set to zero
Reporting	
<u>level</u> (#)	set confidence level; default is <code>level(95)</code>
<u>full</u>	show constrained parameters in table
<u>var</u>	display underlying <code>var</code> output
<u>lutstats</u>	report Lütkepohl lag-order selection statistics
<u>nocnsreport</u>	do not display constraints
<u>display_options</u>	control columns and column formats
Maximization	
<u>maximize_options</u>	control the maximization process; seldom used
<u>coeflegend</u>	display legend instead of statistics
* <code>aconstraints</code> (<i>constraints_a</i>), <code>aeq</code> (<i>matrix_{aeq}</i>), <code>acns</code> (<i>matrix_{acns}</i>), <code>bconstraints</code> (<i>constraints_b</i>), <code>beq</code> (<i>matrix_{beq}</i>), <code>bcns</code> (<i>matrix_{bcns}</i>): at least one of these options must be specified.	
<code>coeflegend</code> does not appear in the dialog box.	

<i>long_run_options</i>	Description
Model	
<u>noconstant</u>	suppress constant term
* <u>lrconstraints</u> (<i>constraints</i> _{lr})	apply previously defined <i>constraints</i> _{lr} to C
* <u>lreq</u> (<i>matrix</i> _{lreq})	define and apply to C equality constraint matrix <i>matrix</i> _{lreq}
* <u>lrcns</u> (<i>matrix</i> _{lrcns})	define and apply to C cross-parameter constraint matrix <i>matrix</i> _{lrcns}
<u>lags</u> (<i>numlist</i>)	use lags <i>numlist</i> in the underlying VAR
Model 2	
<u>exog</u> (<i>varlist</i> _{exog})	use exogenous variables <i>varlist</i>
<u>varconstraints</u> (<i>constraints</i> _v)	apply <i>constraints</i> _v to underlying VAR
<u>noislog</u>	suppress SURE iteration log
<u>isiterate</u> (#)	set maximum number of iterations for SURE; default is <code>isiterate(1600)</code>
<u>istolerance</u> (#)	set convergence tolerance of SURE
<u>noisure</u>	use one-step SURE
<u>dfk</u>	make small-sample degrees-of-freedom adjustment
<u>small</u>	report small-sample <i>t</i> and <i>F</i> statistics
<u>noidencheck</u>	do not check for local identification
<u>nobigf</u>	do not compute parameter vector for coefficients implicitly set to zero
Reporting	
<u>level</u> (#)	set confidence level; default is <code>level(95)</code>
<u>full</u>	show constrained parameters in table
<u>var</u>	display underlying <code>var</code> output
<u>lutstats</u>	report Lütkepohl lag-order selection statistics
<u>nocnsreport</u>	do not display constraints
<u>display_options</u>	control columns and column formats
Maximization	
<u>maximize_options</u>	control the maximization process; seldom used
<u>coeflegend</u>	display legend instead of statistics

* `lrconstraints`(*constraints*_{lr}), `lreq`(*matrix*_{lreq}), `lrcns`(*matrix*_{lrcns}): at least one of these options must be specified.

`coeflegend` does not appear in the dialog box.

You must `tsset` your data before using `svvar`; see [TS] `tsset`.

`depvarlist` and `varlist`_{exog} may contain time-series operators; see [U] 11.4.4 Time-series varlists.

`by`, `fp`, `rolling`, `statsby`, and `xi` are allowed; see [U] 11.1.10 Prefix commands.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Options

Model

noconstant; see [R] estimation options.

aconstraints(*constraints_a*), aeq(*matrix_{aeq}*), acns(*matrix_{acns}*)

bconstraints(*constraints_b*), beq(*matrix_{beq}*), bcns(*matrix_{bcns}*)

These options specify the short-run constraints in an SVAR. To specify a short-run SVAR model, you must specify at least one of these options. The first list of options specifies constraints on the parameters of the **A** matrix; the second list specifies constraints on the parameters of the **B** matrix (see *Short-run SVAR models*). If at least one option is selected from the first list and none are selected from the second list, `svar` sets **B** to the identity matrix. Similarly, if at least one option is selected from the second list and none are selected from the first list, `svar` sets **A** to the identity matrix.

None of these options may be specified with any of the options that define long-run constraints.

aconstraints(*constraints_a*) specifies a *numlist* of previously defined Stata constraints to be applied to **A** during estimation.

aeq(*matrix_{aeq}*) specifies a matrix that defines a set of equality constraints. This matrix must be square with dimension equal to the number of equations in the underlying VAR. The elements of this matrix must be *missing* or real numbers. A missing value in the (*i*, *j*) element of this matrix specifies that the (*i*, *j*) element of **A** is a free parameter. A real number in the (*i*, *j*) element of this matrix constrains the (*i*, *j*) element of **A** to this real number. For example,

$$\mathbf{A} = \begin{bmatrix} 1 & 0 \\ \cdot & 1.5 \end{bmatrix}$$

specifies that $\mathbf{A}[1, 1] = 1$, $\mathbf{A}[1, 2] = 0$, $\mathbf{A}[2, 2] = 1.5$, and $\mathbf{A}[2, 1]$ is a free parameter.

acns(*matrix_{acns}*) specifies a matrix that defines a set of exclusion or cross-parameter equality constraints on **A**. This matrix must be square with dimension equal to the number of equations in the underlying VAR. Each element of this matrix must be *missing*, 0, or a positive integer. A missing value in the (*i*, *j*) element of this matrix specifies that no constraint be placed on this element of **A**. A zero in the (*i*, *j*) element of this matrix constrains the (*i*, *j*) element of **A** to be zero. Any strictly positive integers must be in two or more elements of this matrix. A strictly positive integer in the (*i*, *j*) element of this matrix constrains the (*i*, *j*) element of **A** to be equal to all the other elements of **A** that correspond to elements in this matrix that contain the same integer. For example, consider the matrix

$$\mathbf{A} = \begin{bmatrix} \cdot & 1 \\ 1 & 0 \end{bmatrix}$$

Specifying `acns(A)` in a two-equation SVAR constrains $\mathbf{A}[2, 1] = \mathbf{A}[1, 2]$ and $\mathbf{A}[2, 2] = 0$ while leaving $\mathbf{A}[1, 1]$ free.

bconstraints(*constraints_b*) specifies a *numlist* of previously defined Stata constraints to be applied to **B** during estimation.

beq(*matrix_{beq}*) specifies a matrix that defines a set of equality constraints. This matrix must be square with dimension equal to the number of equations in the underlying VAR. The elements of this matrix must be either *missing* or real numbers. The syntax of implied constraints is analogous to the one described in `aeq()`, except that it applies to **B** rather than to **A**.

`bcns(matrixbcns)` specifies a matrix that defines a set of exclusion or cross-parameter equality constraints on **B**. This matrix must be square with dimension equal to the number of equations in the underlying VAR. Each element of this matrix must be *missing*, 0, or a positive integer. The format of the implied constraints is the same as the one described in the `acns()` option above.

`lrconstraints(constraintslr)`, `lreq(matrixlreq)`, `lrcns(matrixlrcns)`

These options specify the long-run constraints in an SVAR. To specify a long-run SVAR model, you must specify at least one of these options. The list of options specifies constraints on the parameters of the long-run **C** matrix (see *Long-run SVAR models* for the definition of **C**). None of these options may be specified with any of the options that define short-run constraints.

`lrconstraints(constraintslr)` specifies a *numlist* of previously defined Stata constraints to be applied to **C** during estimation.

`lreq(matrixlreq)` specifies a matrix that defines a set of equality constraints on the elements of **C**. This matrix must be square with dimension equal to the number of equations in the underlying VAR. The elements of this matrix must be either *missing* or real numbers. The syntax of implied constraints is analogous to the one described in option `aeq()`, except that it applies to **C**.

`lrcns(matrixlrcns)` specifies a matrix that defines a set of exclusion or cross-parameter equality constraints on **C**. This matrix must be square with dimension equal to the number of equations in the underlying VAR. Each element of this matrix must be *missing*, 0, or a positive integer. The syntax of the implied constraints is the same as the one described for the `acns()` option above.

`lags(numlist)` specifies the lags to be included in the underlying VAR model. The default is `lags(1 2)`. This option takes a *numlist* and not simply an integer for the maximum lag. For instance, `lags(2)` would include only the second lag in the model, whereas `lags(1/2)` would include both the first and second lags in the model. See [U] 11.1.8 *numlist* and [U] 11.4.4 *Time-series varlists* for further discussion of *numlists* and lags.

Model 2

`exog(varlistexog)` specifies a list of exogenous variables to be included in the underlying VAR.

`varconstraints(constraintsv)` specifies a list of constraints to be applied to the coefficients in the underlying VAR. Because `svar` estimates multiple equations, the constraints must specify the equation name for all but the first equation.

`noislog` prevents `svar` from displaying the iteration log from the iterated seemingly unrelated regression algorithm. When the `varconstraints()` option is not specified, the VAR coefficients are estimated via OLS, a noniterative procedure. As a result, `noislog` may be specified only with `varconstraints()`. Similarly, `noislog` may not be combined with `noisure`.

`isiterate(#)` sets the maximum number of iterations for the iterated seemingly unrelated regression algorithm. The default limit is 1,600. When the `varconstraints()` option is not specified, the VAR coefficients are estimated via OLS, a noniterative procedure. As a result, `isiterate()` may be specified only with `varconstraints()`. Similarly, `isiterate()` may not be combined with `noisure`.

`istolerance(#)` specifies the convergence tolerance of the iterated seemingly unrelated regression algorithm. The default tolerance is `1e-6`. When the `varconstraints()` option is not specified, the VAR coefficients are estimated via OLS, a noniterative procedure. As a result, `istolerance()` may be specified only with `varconstraints()`. Similarly, `istolerance()` may not be combined with `noisure`.

`noisure` specifies that the VAR coefficients be estimated via one-step seemingly unrelated regression when `varconstraints()` is specified. By default, `svvar` estimates the coefficients in the VAR via iterated seemingly unrelated regression when `varconstraints()` is specified. When the `varconstraints()` option is not specified, the VAR coefficient estimates are obtained via OLS, a noniterative procedure. As a result, `noisure` may be specified only with `varconstraints()`.

`dfk` specifies that a small-sample degrees-of-freedom adjustment be used when estimating Σ , the covariance matrix of the VAR disturbances. Specifically, $1/(T - \bar{m})$ is used instead of the large-sample divisor $1/T$, where \bar{m} is the average number of parameters in the functional form for y_t over the K equations.

`small` causes `svvar` to calculate and report small-sample t and F statistics instead of the large-sample normal and chi-squared statistics.

`noidencheck` requests that the [Amisano and Giannini \(1997\)](#) check for local identification not be performed. This check is local to the starting values used. Because of this dependence on the starting values, you may wish to suppress this check by specifying the `noidencheck` option. However, be careful in specifying this option. Models that are not structurally identified can still converge, thereby producing meaningless results that only appear to have meaning.

`nobigf` requests that `svvar` not compute the estimated parameter vector that incorporates coefficients that have been implicitly constrained to be zero, such as when some lags have been omitted from a model. `e(bf)` is used for computing asymptotic standard errors in the postestimation commands `irf create` and `fcast compute`. Therefore, specifying `nobigf` implies that the asymptotic standard errors will not be available from `irf create` and `fcast compute`. See [Fitting models with some lags excluded](#) in [TS] `var`.

Reporting

`level(#)`; see [R] [estimation options](#).

`full` shows constrained parameters in table.

`var` specifies that the output from `var` also be displayed. By default, the underlying VAR is fit quietly.

`lutstats` specifies that the Lütkepohl versions of the lag-order selection statistics be computed. See [Methods and formulas](#) in [TS] `varsoc` for a discussion of these statistics.

`nocnsreport`; see [R] [estimation options](#).

`display_options`: `nocl`, `nopvalues`, `cformat(%fmt)`, `pformat(%fmt)`, and `sformat(%fmt)`; see [R] [estimation options](#).

Maximization

`maximize_options`: `difficult`, `technique(algorithm_spec)`, `iterate(#)`, `[no]log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(#)`, `ltolerance(#)`, `nrtolerance(#)`, `nonrtolerance`, and `from(init_specs)`; see [R] [maximize](#). These options are seldom used.

The following option is available with `svvar` but is not shown in the dialog box:

`coeflegend`; see [R] [estimation options](#).

Remarks and examples

Remarks are presented under the following headings:

[Introduction](#)
[Short-run SVAR models](#)
[Long-run SVAR models](#)

Introduction

This entry assumes that you have already read [TS] [var intro](#) and [TS] [var](#); if not, please do. Here we illustrate how to fit SVARs in Stata subject to short-run and long-run restrictions. For more detailed information on SVARs, see [Amisano and Giannini \(1997\)](#) and [Hamilton \(1994\)](#). For good introductions to VARs, see [Lütkepohl \(2005\)](#), [Hamilton \(1994\)](#), [Stock and Watson \(2001\)](#), and [Becketti \(2013\)](#).

Short-run SVAR models

A short-run SVAR model without exogenous variables can be written as

$$\mathbf{A}(\mathbf{I}_K - \mathbf{A}_1L - \mathbf{A}_2L^2 - \dots - \mathbf{A}_pL^p)\mathbf{y}_t = \mathbf{A}\boldsymbol{\epsilon}_t = \mathbf{B}\mathbf{e}_t$$

where L is the lag operator, \mathbf{A} , \mathbf{B} , and $\mathbf{A}_1, \dots, \mathbf{A}_p$ are $K \times K$ matrices of parameters, $\boldsymbol{\epsilon}_t$ is a $K \times 1$ vector of innovations with $\boldsymbol{\epsilon}_t \sim N(\mathbf{0}, \boldsymbol{\Sigma})$ and $E[\boldsymbol{\epsilon}_t\boldsymbol{\epsilon}_s'] = \mathbf{0}_K$ for all $s \neq t$, and \mathbf{e}_t is a $K \times 1$ vector of orthogonalized disturbances; that is, $\mathbf{e}_t \sim N(\mathbf{0}, \mathbf{I}_K)$ and $E[\mathbf{e}_t\mathbf{e}_s'] = \mathbf{0}_K$ for all $s \neq t$. These transformations of the innovations allow us to analyze the dynamics of the system in terms of a change to an element of \mathbf{e}_t . In a short-run SVAR model, we obtain identification by placing restrictions on \mathbf{A} and \mathbf{B} , which are assumed to be nonsingular.

► Example 1: Short-run just-identified SVAR model

Following [Sims \(1980\)](#), the Cholesky decomposition is one method of identifying the impulse–response functions in a VAR; thus, this method corresponds to an SVAR. There are several sets of constraints on \mathbf{A} and \mathbf{B} that are easily manipulated back to the Cholesky decomposition, and the following example illustrates this point.

One way to impose the Cholesky restrictions is to assume an SVAR model of the form

$$\tilde{\mathbf{A}}(\mathbf{I}_K - \mathbf{A}_1 - \mathbf{A}_2L^2 - \dots - \mathbf{A}_pL^p)\mathbf{y}_t = \tilde{\mathbf{B}}\mathbf{e}_t$$

where $\tilde{\mathbf{A}}$ is a lower triangular matrix with ones on the diagonal and $\tilde{\mathbf{B}}$ is a diagonal matrix. Because the \mathbf{P} matrix for this model is $\mathbf{P}_{\text{sr}} = \tilde{\mathbf{A}}^{-1}\tilde{\mathbf{B}}$, its estimate, $\hat{\mathbf{P}}_{\text{sr}}$, obtained by plugging in estimates of $\tilde{\mathbf{A}}$ and $\tilde{\mathbf{B}}$, should equal the Cholesky decomposition of $\hat{\boldsymbol{\Sigma}}$.

To illustrate, we use the German macroeconomic data discussed in [Lütkepohl \(2005\)](#) and used in [TS] [var](#). In this example, $\mathbf{y}_t = (\text{dln_inv}, \text{dln_inc}, \text{dln_consump})$, where `dln_inv` is the first difference of the log of investment, `dln_inc` is the first difference of the log of income, and `dln_consump` is the first difference of the log of consumption. Because the first difference of the natural log of a variable can be treated as an approximation of the percentage change in that variable, we will refer to these variables as percentage changes in `inv`, `inc`, and `consump`, respectively.

We will impose the Cholesky restrictions on this system by applying equality constraints with the constraint matrices

$$\mathbf{A} = \begin{bmatrix} 1 & 0 & 0 \\ . & 1 & 0 \\ . & . & 1 \end{bmatrix} \quad \text{and} \quad \mathbf{B} = \begin{bmatrix} . & 0 & 0 \\ 0 & . & 0 \\ 0 & 0 & . \end{bmatrix}$$

With these structural restrictions, we assume that the percentage change in `inv` is not contemporaneously affected by the percentage changes in either `inc` or `consump`. We also assume that the percentage change of `inc` is affected by contemporaneous changes in `inv` but not `consump`. Finally, we assume that percentage changes in `consump` are affected by contemporaneous changes in both `inv` and `inc`.

The following commands fit an SVAR model with these constraints.

```
. use http://www.stata-press.com/data/r14/lutkepohl2
(Quarterly SA West German macro data, Bil DM, from Lutkepohl 1993 Table E.1)
. matrix A = (1,0,0\.,1,0\.,.,1)
. matrix B = (.,0,0\0,.,0\0,0,.)
. svar dln_inv dln_inc dln_consump if qtr<=tq(1978q4), aeq(A) beq(B)
Estimating short-run parameters
```

(output omitted)

Structural vector autoregression

```
( 1) [a_1_1]_cons = 1
( 2) [a_1_2]_cons = 0
( 3) [a_1_3]_cons = 0
( 4) [a_2_2]_cons = 1
( 5) [a_2_3]_cons = 0
( 6) [a_3_3]_cons = 1
( 7) [b_1_2]_cons = 0
( 8) [b_1_3]_cons = 0
( 9) [b_2_1]_cons = 0
(10) [b_2_3]_cons = 0
(11) [b_3_1]_cons = 0
(12) [b_3_2]_cons = 0
```

Sample: 1960q4 - 1978q4
Exactly identified model

Number of obs = 73
Log likelihood = 606.307

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
/a_1_1	1	(constrained)				
/a_2_1	-.0336288	.0294605	-1.14	0.254	-.0913702	.0241126
/a_3_1	-.0435846	.0194408	-2.24	0.025	-.0816879	-.0054812
/a_1_2	0	(constrained)				
/a_2_2	1	(constrained)				
/a_3_2	-.424774	.0765548	-5.55	0.000	-.5748187	-.2747293
/a_1_3	0	(constrained)				
/a_2_3	0	(constrained)				
/a_3_3	1	(constrained)				
/b_1_1	.0438796	.0036315	12.08	0.000	.036762	.0509972
/b_2_1	0	(constrained)				
/b_3_1	0	(constrained)				
/b_1_2	0	(constrained)				
/b_2_2	.0110449	.0009141	12.08	0.000	.0092534	.0128365
/b_3_2	0	(constrained)				
/b_1_3	0	(constrained)				
/b_2_3	0	(constrained)				
/b_3_3	.0072243	.0005979	12.08	0.000	.0060525	.0083962

The SVAR output has four parts: an iteration log, a display of the constraints imposed, a header with sample and SVAR log-likelihood information, and a table displaying the estimates of the parameters from the **A** and **B** matrices. From the output above, we can see that the equality constraint matrices supplied to `svar` imposed the intended constraints and that the SVAR header informs us that the model we fit is just identified. The estimates of `a_2_1`, `a_3_1`, and `a_3_2` are all negative. Because the

off-diagonal elements of the \mathbf{A} matrix contain the negative of the actual contemporaneous effects, the estimated effects are positive, as expected.

The estimates $\widehat{\mathbf{A}}$ and $\widehat{\mathbf{B}}$ are stored in `e(A)` and `e(B)`, respectively, allowing us to compute the estimated Cholesky decomposition.

```
. matrix Aest = e(A)
. matrix Best = e(B)
. matrix chol_est = inv(Aest)*Best
. matrix list chol_est
chol_est[3,3]
      dln_inv      dln_inc      dln_consump
dln_inv      .04387957      0      0
dln_inc      .00147562      .01104494      0
dln_consump  .00253928      .0046916      .00722432
```

`svar` stores the estimated Σ from the underlying `var` in `e(Sigma)`. The output below illustrates the computation of the Cholesky decomposition of `e(Sigma)`. It is the same as the output computed from the SVAR estimates.

```
. matrix sig_var = e(Sigma)
. matrix chol_var = cholesky(sig_var)
. matrix list chol_var
chol_var[3,3]
      dln_inv      dln_inc      dln_consump
dln_inv      .04387957      0      0
dln_inc      .00147562      .01104494      0
dln_consump  .00253928      .0046916      .00722432
```

◀

We might now wonder why we bother obtaining parameter estimates via nonlinear estimation if we can obtain them simply by a transform of the estimates produced by `var`. When the model is just identified, as in the [previous example](#), the SVAR parameter estimates can be computed via a transform of the VAR estimates. However, when the model is overidentified, such is not the case.

▶ Example 2: Short-run overidentified SVAR model

The Cholesky decomposition example above fit a just-identified model. This example considers an overidentified model. In [example 1](#), the `a_2_1` parameter was not significant, which is consistent with a theory in which changes in our measure of investment affect only changes in income with a lag. We can impose the restriction that `a_2_1` is zero and then test this overidentifying restriction. Our \mathbf{A} and \mathbf{B} matrices are now

$$\mathbf{A} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ . & . & 1 \end{bmatrix} \quad \text{and} \quad \mathbf{B} = \begin{bmatrix} . & 0 & 0 \\ 0 & . & 0 \\ 0 & 0 & . \end{bmatrix}$$

The output below contains the commands and results we obtained by fitting this model on the Lütkepohl data.

```
. matrix B = (.,0,0\0,.,.,0\0,0,.)
. matrix A = (1,0,0\0,1,0\.,.,1)
```

```
. svar dln_inv dln_inc dln_consump if qtr<=tq(1978q4), aeq(A) beq(B)
Estimating short-run parameters
(output omitted)
```

```
Structural vector autoregression
```

```
( 1) [a_1_1]_cons = 1
( 2) [a_1_2]_cons = 0
( 3) [a_1_3]_cons = 0
( 4) [a_2_1]_cons = 0
( 5) [a_2_2]_cons = 1
( 6) [a_2_3]_cons = 0
( 7) [a_3_3]_cons = 1
( 8) [b_1_2]_cons = 0
( 9) [b_1_3]_cons = 0
(10) [b_2_1]_cons = 0
(11) [b_2_3]_cons = 0
(12) [b_3_1]_cons = 0
(13) [b_3_2]_cons = 0
```

```
Sample: 1960q4 - 1978q4
Overidentified model
```

```
Number of obs   =      73
Log likelihood   =  605.6613
```

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
/a_1_1	1	(constrained)				
/a_2_1	0	(constrained)				
/a_3_1	-.0435911	.0192696	-2.26	0.024	-.0813589	-.0058233
/a_1_2	0	(constrained)				
/a_2_2	1	(constrained)				
/a_3_2	-.4247741	.0758806	-5.60	0.000	-.5734973	-.2760508
/a_1_3	0	(constrained)				
/a_2_3	0	(constrained)				
/a_3_3	1	(constrained)				
/b_1_1	.0438796	.0036315	12.08	0.000	.036762	.0509972
/b_2_1	0	(constrained)				
/b_3_1	0	(constrained)				
/b_1_2	0	(constrained)				
/b_2_2	.0111431	.0009222	12.08	0.000	.0093356	.0129506
/b_3_2	0	(constrained)				
/b_1_3	0	(constrained)				
/b_2_3	0	(constrained)				
/b_3_3	.0072243	.0005979	12.08	0.000	.0060525	.0083962

```
LR test of identifying restrictions: chi2(1) = 1.292      Prob > chi2 = 0.256
```

The footer in this example reports a test of the overidentifying restriction. The null hypothesis of this test is that any overidentifying restrictions are valid. In the case at hand, we cannot reject this null hypothesis at any of the conventional levels.

◀

► Example 3: Short-run SVAR model with constraints

`svvar` also allows us to place constraints on the parameters of the underlying VAR. We begin by looking at the underlying VAR for the SVARs that we have used in the previous examples.

```
. var dln_inv dln_inc dln_consump if qtr<=tq(1978q4)
```

```
Vector autoregression
```

```
Sample: 1960q4 - 1978q4          Number of obs   =          73
Log likelihood = 606.307          AIC              = -16.03581
FPE            = 2.18e-11         HQIC             = -15.77323
Det(Sigma_ml) = 1.23e-11         SBIC             = -15.37691
```

Equation	Parms	RMSE	R-sq	chi2	P>chi2
dln_inv	7	.046148	0.1286	10.76961	0.0958
dln_inc	7	.011719	0.1142	9.410683	0.1518
dln_consump	7	.009445	0.2513	24.50031	0.0004

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
dln_inv						
dln_inv						
L1.	-.3196318	.1192898	-2.68	0.007	-.5534355	-.0858282
L2.	-.1605508	.118767	-1.35	0.176	-.39333	.0722283
dln_inc						
L1.	.1459851	.5188451	0.28	0.778	-.8709326	1.162903
L2.	.1146009	.508295	0.23	0.822	-.881639	1.110841
dln_consump						
L1.	.9612288	.6316557	1.52	0.128	-.2767936	2.199251
L2.	.9344001	.6324034	1.48	0.140	-.3050877	2.173888
_cons	-.0167221	.0163796	-1.02	0.307	-.0488257	.0153814
dln_inc						
dln_inv						
L1.	.0439309	.0302933	1.45	0.147	-.0154427	.1033046
L2.	.0500302	.0301605	1.66	0.097	-.0090833	.1091437
dln_inc						
L1.	-.1527311	.131759	-1.16	0.246	-.4109741	.1055118
L2.	.0191634	.1290799	0.15	0.882	-.2338285	.2721552
dln_consump						
L1.	.2884992	.1604069	1.80	0.072	-.0258926	.6028909
L2.	-.0102	.1605968	-0.06	0.949	-.3249639	.3045639
_cons	.0157672	.0041596	3.79	0.000	.0076146	.0239198
dln_consump						
dln_inv						
L1.	-.002423	.0244142	-0.10	0.921	-.050274	.045428
L2.	.0338806	.0243072	1.39	0.163	-.0137607	.0815219
dln_inc						
L1.	.2248134	.1061884	2.12	0.034	.0166879	.4329389
L2.	.3549135	.1040292	3.41	0.001	.1510199	.558807
dln_consump						
L1.	-.2639695	.1292766	-2.04	0.041	-.517347	-.010592
L2.	-.0222264	.1294296	-0.17	0.864	-.2759039	.231451
_cons	.0129258	.0033523	3.86	0.000	.0063554	.0194962

The equation-level model tests reported in the header indicate that we cannot reject the null hypotheses that all the coefficients in the first equation are zero, nor can we reject the null that all the coefficients in the second equation are zero at the 5% significance level. We use a combination of theory and the p -values from the output above to place some exclusion restrictions on the underlying VAR(2). Specifically, in the equation for the percentage change of `inv`, we constrain the coefficients on `L2.dln_inv`, `L.dln_inc`, `L2.dln_inc`, and `L2.dln_consump` to be zero. In the equation for `dln_inc`, we constrain the coefficients on `L2.dln_inv`, `L2.dln_inc`, and `L2.dln_consump` to be zero. Finally, in the equation for `dln_consump`, we constrain `L.dln_inv` and `L2.dln_consump` to be zero. We then refit the SVAR from the [previous example](#).

```
. constraint 1 [dln_inv]L2.dln_inv = 0
. constraint 2 [dln_inv ]L.dln_inc = 0
. constraint 3 [dln_inv]L2.dln_inc = 0
. constraint 4 [dln_inv]L2.dln_consump = 0
. constraint 5 [dln_inc]L2.dln_inv = 0
. constraint 6 [dln_inc]L2.dln_inc = 0
. constraint 7 [dln_inc]L2.dln_consump = 0
. constraint 8 [dln_consump]L.dln_inv = 0
. constraint 9 [dln_consump]L2.dln_consump = 0
. svar dln_inv dln_inc dln_consump if qtr<=tq(1978q4), aeq(A) beq(B)
> varconst(1/9) noislog
```

Estimating short-run parameters

(output omitted)

Structural vector autoregression

```
( 1) [a_1_1]_cons = 1
( 2) [a_1_2]_cons = 0
( 3) [a_1_3]_cons = 0
( 4) [a_2_1]_cons = 0
( 5) [a_2_2]_cons = 1
( 6) [a_2_3]_cons = 0
( 7) [a_3_3]_cons = 1
( 8) [b_1_2]_cons = 0
( 9) [b_1_3]_cons = 0
(10) [b_2_1]_cons = 0
(11) [b_2_3]_cons = 0
(12) [b_3_1]_cons = 0
(13) [b_3_2]_cons = 0
```


▷ Example 4: Long-run SVAR model

Suppose that we have a theory in which unexpected changes to the money supply have no long-run effects on changes in output and, similarly, that unexpected changes in output have no long-run effects on changes in the money supply. The \mathbf{C} matrix implied by this theory is

$$\mathbf{C} = \begin{bmatrix} \cdot & 0 \\ 0 & \cdot \end{bmatrix}$$

```
. use http://www.stata-press.com/data/r14/m1gdp
. matrix lr = (.,0\0,.)
. svar d.ln_m1 d.ln_gdp, lreq(lr)
Estimating long-run parameters
(output omitted)
Structural vector autoregression
( 1) [c_1_2]_cons = 0
( 2) [c_2_1]_cons = 0
Sample: 1959q4 - 2002q2                Number of obs   =       171
Overidentified model                    Log likelihood   =    1151.614
```

	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
/c_1_1	.0301007	.0016277	18.49	0.000	.0269106	.0332909
/c_2_1	0	(constrained)				
/c_1_2	0	(constrained)				
/c_2_2	.0129691	.0007013	18.49	0.000	.0115946	.0143436

```
LR test of identifying restrictions: chi2(1) = .1368      Prob > chi2 = 0.712
```

We have assumed that the underlying VAR has 2 lags; four of the five selection-order criteria computed by varsoc (see [TS] varsoc) recommended this choice. The test of the overidentifying restrictions provides no indication that it is not valid.

Stored results

svvar stores the following in `e()`:

Scalars

<code>e(N)</code>	number of observations
<code>e(N_cns)</code>	number of constraints
<code>e(k_eq)</code>	number of equations in <code>e(b)</code>
<code>e(k_dv)</code>	number of dependent variables
<code>e(k_aux)</code>	number of auxiliary parameters
<code>e(ll)</code>	log likelihood from svar
<code>e(ll_#)</code>	log likelihood for equation #
<code>e(N_gaps_var)</code>	number of gaps in the sample
<code>e(k_var)</code>	number of coefficients in VAR
<code>e(k_eq_var)</code>	number of equations in underlying VAR
<code>e(k_dv_var)</code>	number of dependent variables in underlying VAR
<code>e(df_eq_var)</code>	average number of parameters in an equation
<code>e(df_m_var)</code>	model degrees of freedom
<code>e(df_r_var)</code>	if <code>small</code> , residual degrees of freedom
<code>e(obs_#_var)</code>	number of observations on equation #
<code>e(k_#_var)</code>	number of coefficients in equation #
<code>e(df_m#_var)</code>	model degrees of freedom for equation #
<code>e(df_r#_var)</code>	residual degrees of freedom for equation # (<code>small</code> only)
<code>e(r2_#_var)</code>	R -squared for equation #
<code>e(ll_#_var)</code>	log likelihood for equation # VAR
<code>e(chi2_#_var)</code>	χ^2 statistic for equation #
<code>e(F_#_var)</code>	F statistic for equation # (<code>small</code> only)
<code>e(rmse_#_var)</code>	root mean squared error for equation #
<code>e(mlag_var)</code>	highest lag in VAR
<code>e(tparms_var)</code>	number of parameters in all equations
<code>e(aic_var)</code>	Akaike information criterion
<code>e(hqic_var)</code>	Hannan–Quinn information criterion
<code>e(sbic_var)</code>	Schwarz–Bayesian information criterion
<code>e(fpe_var)</code>	final prediction error
<code>e(ll_var)</code>	log likelihood from var
<code>e(detsig_var)</code>	determinant of <code>e(Sigma)</code>
<code>e(detsig_ml_var)</code>	determinant of $\widehat{\Sigma}_{ml}$
<code>e(tmin)</code>	first time period in the sample
<code>e(tmax)</code>	maximum time
<code>e(chi2_oid)</code>	overidentification test
<code>e(oid_df)</code>	number of overidentifying restrictions
<code>e(rank)</code>	rank of <code>e(V)</code>
<code>e(ic_ml)</code>	number of iterations
<code>e(rc_ml)</code>	return code from ml

Macros

e(cmd)	svar
e(cmdline)	command as typed
e(lrmodel)	long-run model, if specified
e(lags_var)	lags in model
e(depvar_var)	names of dependent variables
e(endog_var)	names of endogenous variables
e(exog_var)	names of exogenous variables, if specified
e(nocons_var)	noconstant, if noconstant specified
e(cns_lr)	long-run constraints
e(cns_a)	cross-parameter equality constraints on A
e(cns_b)	cross-parameter equality constraints on B
e(dfk_var)	alternate divisor (dfk), if specified
e(eqnames_var)	names of equations
e(lutstats_var)	lutstats, if specified
e(constraints_var)	constraints_var, if there are constraints on VAR
e(small)	small, if specified
e(tsfmt)	format of timevar
e(timevar)	name of timevar
e(title)	title in estimation output
e(properties)	b V
e(predict)	program used to implement predict

Matrices

e(b)	coefficient vector
e(Cns)	constraints matrix
e(Sigma)	$\hat{\Sigma}$ matrix
e(V)	variance-covariance matrix of the estimators
e(b_var)	coefficient vector of underlying VAR model
e(V_var)	VCE of underlying VAR model
e(bf_var)	full coefficient vector with zeros in dropped lags
e(G_var)	G matrix stored by var; see [TSS] var Methods and formulas
e(aeq)	aeq(matrix), if specified
e(acns)	acns(matrix), if specified
e(beq)	beq(matrix), if specified
e(bcns)	bcns(matrix), if specified
e(lreq)	lreq(matrix), if specified
e(lrcns)	lrcns(matrix), if specified
e(Cns_var)	constraint matrix from var, if varconstraints() is specified
e(A)	estimated A matrix, if a short-run model
e(B)	estimated B matrix
e(C)	estimated C matrix, if a long-run model
e(A1)	estimated \bar{A} matrix, if a long-run model

Functions

e(sample)	marks estimation sample
-----------	-------------------------

Methods and formulas

The log-likelihood function for models with short-run constraints is

$$L(\mathbf{A}, \mathbf{B}) = -\frac{NK}{2} \ln(2\pi) + \frac{N}{2} \ln(|\mathbf{W}|^2) - \frac{N}{2} \text{tr}(\mathbf{W}'\mathbf{W}\hat{\Sigma})$$

where $\mathbf{W} = \mathbf{B}^{-1}\mathbf{A}$.

When there are long-run constraints, because $\mathbf{C} = \bar{\mathbf{A}}^{-1}\mathbf{B}$ and $\mathbf{A} = \mathbf{I}_K$, $\mathbf{W} = \mathbf{B}^{-1} = \mathbf{C}^{-1}\bar{\mathbf{A}}^{-1} = (\bar{\mathbf{A}}\mathbf{C})^{-1}$. Substituting the last term for \mathbf{W} in the short-run log likelihood produces the long-run log likelihood

$$L(\mathbf{C}) = -\frac{NK}{2} \ln(2\pi) + \frac{N}{2} \ln(|\widetilde{\mathbf{W}}|^2) - \frac{N}{2} \text{tr}(\widetilde{\mathbf{W}}' \widetilde{\mathbf{W}} \widehat{\boldsymbol{\Sigma}})$$

where $\widetilde{\mathbf{W}} = (\bar{\mathbf{A}}\mathbf{C})^{-1}$.

For both the short-run and the long-run models, the maximization is performed by the scoring method. See [Harvey \(1990\)](#) for a discussion of this method.

Based on results from [Amisano and Giannini \(1997\)](#), the score vector for the short-run model is

$$\frac{\partial L(\mathbf{A}, \mathbf{B})}{\partial [\text{vec}(\mathbf{A}), \text{vec}(\mathbf{B})]} = N \left[\{\text{vec}(\mathbf{W}'^{-1})\}' - \{\text{vec}(\mathbf{W})\}' (\widehat{\boldsymbol{\Sigma}} \otimes \mathbf{I}_K) \right] \times \\ [(\mathbf{I}_K \otimes \mathbf{B}^{-1}), -(\mathbf{A}'\mathbf{B}'^{-1} \otimes \mathbf{B}^{-1})]$$

and the expected information matrix is

$$I[\text{vec}(\mathbf{A}), \text{vec}(\mathbf{B})] = N \left[\begin{array}{c} (\mathbf{W}^{-1} \otimes \mathbf{B}'^{-1}) \\ -(\mathbf{I}_K \otimes \mathbf{B}'^{-1}) \end{array} \right] (\mathbf{I}_{K^2} + \oplus) [(\mathbf{W}'^{-1} \otimes \mathbf{B}^{-1}), -(\mathbf{I}_K \otimes \mathbf{B}^{-1})]$$

where \oplus is the commutation matrix defined in [Magnus and Neudecker \(1999, 46–48\)](#).

Using results from [Amisano and Giannini \(1997\)](#), we can derive the score vector and the expected information matrix for the case with long-run restrictions. The score vector is

$$\frac{\partial L(\mathbf{C})}{\partial \text{vec}(\mathbf{C})} = N \left[\{\text{vec}(\mathbf{W}'^{-1})\}' - \{\text{vec}(\mathbf{W})\}' (\widehat{\boldsymbol{\Sigma}} \otimes \mathbf{I}_K) \right] [-(\bar{\mathbf{A}}'^{-1} \mathbf{C}'^{-1} \otimes \mathbf{C}^{-1})]$$

and the expected information matrix is

$$I[\text{vec}(\mathbf{C})] = N(\mathbf{I}_K \otimes \mathbf{C}'^{-1})(\mathbf{I}_{K^2} + \oplus)(\mathbf{I}_K \otimes \mathbf{C}'^{-1})$$

Checking for identification

This section describes the methods used to check for identification of models with short-run or long-run constraints. Both methods depend on the starting values. By default, `sva`r uses starting values constructed by taking a vector of appropriate dimension and applying the constraints. If there are m parameters in the model, the j th element of the $1 \times m$ vector is $1 + m/100$. `sva`r also allows the user to provide starting values.

For the short-run case, the model is identified if the matrix

$$\mathbf{V}_{\text{sr}}^* = \left[\begin{array}{cc} \mathbf{N}_K & \mathbf{N}_K \\ \mathbf{N}_K & \mathbf{N}_K \\ \mathbf{R}_a(\mathbf{W}' \otimes \mathbf{B}) & \mathbf{0}_{K^2} \\ \mathbf{0}_{K^2} & \mathbf{R}_a(\mathbf{I}_K \otimes \mathbf{B}) \end{array} \right]$$

has full column rank of $2K^2$, where $\mathbf{N}_K = (1/2)(\mathbf{I}_{K^2} + \oplus)$, \mathbf{R}_a is the constraint matrix for the parameters in \mathbf{A} (that is, $\mathbf{R}_a \text{vec}(\mathbf{A}) = \mathbf{r}_a$), and \mathbf{R}_b is the constraint matrix for the parameters in \mathbf{B} (that is, $\mathbf{R}_b \text{vec}(\mathbf{B}) = \mathbf{r}_b$).

For the long-run case, based on results from the \mathbf{C} model in Amisano and Giannini (1997), the model is identified if the matrix

$$\mathbf{V}_{\text{lr}}^* = \begin{bmatrix} (\mathbf{I} \otimes \mathbf{C}'^{-1})(2\mathbf{N}_K)(\mathbf{I} \otimes \mathbf{C}^{-1}) \\ \mathbf{R}_c \end{bmatrix}$$

has full column rank of K^2 , where \mathbf{R}_c is the constraint matrix for the parameters in \mathbf{C} ; that is, $\mathbf{R}_c \text{vec}(\mathbf{C}) = \mathbf{r}_c$.

The test of the overidentifying restrictions is computed as

$$LR = 2(LL_{\text{var}} - LL_{\text{svar}})$$

where LR is the value of the test statistic against the null hypothesis that the overidentifying restrictions are valid, LL_{var} is the log likelihood from the underlying VAR(p) model, and LL_{svar} is the log likelihood from the SVAR model. The test statistic is asymptotically distributed as $\chi^2(q)$, where q is the number of overidentifying restrictions. Amisano and Giannini (1997, 38–39) emphasize that, because this test of the validity of the overidentifying restrictions is an omnibus test, it can be interpreted as a test of the null hypothesis that all the restrictions are valid.

Because constraints might not be independent either by construction or because of the data, the number of restrictions is not necessarily equal to the number of constraints. The rank of $\mathbf{e}(\mathbf{V})$ gives the number of parameters that were independently estimated after applying the constraints. The maximum number of parameters that can be estimated in an identified short-run or long-run SVAR is $K(K+1)/2$. This implies that the number of overidentifying restrictions, q , is equal to $K(K+1)/2$ minus the rank of $\mathbf{e}(\mathbf{V})$.

The number of overidentifying restrictions is also linked to the order condition for each model. In a short-run SVAR model, there are $2K^2$ parameters. Because no more than $K(K+1)/2$ parameters may be estimated, the order condition for a short-run SVAR model is that at least $2K^2 - K(K+1)/2$ restrictions be placed on the model. Similarly, there are K^2 parameters in long-run SVAR model. Because no more than $K(K+1)/2$ parameters may be estimated, the order condition for a long-run SVAR model is that at least $K^2 - K(K+1)/2$ restrictions be placed on the model.

Acknowledgment

We thank Gianni Amisano of the Dipartimento di Scienze Economiche at the Università degli Studi di Brescia for his helpful comments.

References

- Amisano, G., and C. Giannini. 1997. *Topics in Structural VAR Econometrics*. 2nd ed. Heidelberg: Springer.
- Beckett, S. 2013. *Introduction to Time Series Using Stata*. College Station, TX: Stata Press.
- Christiano, L. J., M. Eichenbaum, and C. L. Evans. 1999. Monetary policy shocks: What have we learned and to what end? In *Handbook of Macroeconomics: Volume 1A*, ed. J. B. Taylor and M. Woodford. New York: Elsevier.
- Hamilton, J. D. 1994. *Time Series Analysis*. Princeton: Princeton University Press.
- Harvey, A. C. 1990. *The Econometric Analysis of Time Series*. 2nd ed. Cambridge, MA: MIT Press.
- Lütkepohl, H. 1993. *Introduction to Multiple Time Series Analysis*. 2nd ed. New York: Springer.
- . 2005. *New Introduction to Multiple Time Series Analysis*. New York: Springer.
- Magnus, J. R., and H. Neudecker. 1999. *Matrix Differential Calculus with Applications in Statistics and Econometrics*. Rev. ed. New York: Wiley.

- Rothenberg, T. J. 1971. Identification in parametric models. *Econometrica* 39: 577–591.
- Sims, C. A. 1980. Macroeconomics and reality. *Econometrica* 48: 1–48.
- Stock, J. H., and M. W. Watson. 2001. Vector autoregressions. *Journal of Economic Perspectives* 15: 101–115.
- Watson, M. W. 1994. Vector autoregressions and cointegration. In Vol. 4 of *Handbook of Econometrics*, ed. R. F. Engle and D. L. McFadden. Amsterdam: Elsevier.

Also see

- [TS] [var svar postestimation](#) — Postestimation tools for svar
- [TS] [tsset](#) — Declare data to be time-series data
- [TS] [var](#) — Vector autoregressive models
- [TS] [varbasic](#) — Fit a simple VAR and graph IRFs or FEVDs
- [TS] [vec](#) — Vector error-correction models
- [U] [20 Estimation and postestimation commands](#)
- [TS] [var intro](#) — Introduction to vector autoregressive models